

Creating WIM images of Windows for system deployment using Windows Vista PE 2.x

10/17/2008
Isaac Holmes
University of Notre Dame

Updated 6/3/2010

Table of Contents

| | |
|---|----|
| Introduction: | 4 |
| Requirements:..... | 5 |
| Creating a basic Windows PE disk for troubleshooting or image capture or deployment:..... | 6 |
| Installing Drivers | 8 |
| Installing drivers to offline Windows PE image: | 8 |
| Installing drivers to running Windows PE:..... | 8 |
| Capturing a WIM disk images using Windows PE and ImageX:..... | 9 |
| Capture to USB or local drive:..... | 9 |
| Capture to a network file share: | 10 |
| Partition and Format the Hard Drive from Windows PE: | 11 |
| Partition and Format drive from Windows PE Single Partition: | 11 |
| Partition and Format drive from Windows PE Two Partitions: | 12 |
| DISKPART Scripting:..... | 13 |
| Deploying WIM images:..... | 14 |
| Deploy from USB drive:..... | 14 |
| Deploy from Network Share: | 14 |
| Deploy from CD/DVD: | 15 |
| Command Files for automation of Base Windows PE ISO Image: | 16 |
| Create_Mount.cmd..... | 16 |
| Install_Packages.cmd | 16 |
| Install_Drivers.CMD | 16 |
| Commit_Changes.cmd | 17 |
| Create_ISO.CMD | 17 |
| Build_All.CMD | 17 |
| Further Customizations:..... | 19 |

| | |
|--|----|
| Adding your own startup commands or applications..... | 19 |
| Remove boot prompt..... | 19 |
| Using Ghost instead of ImageX..... | 19 |
| Modify the WinPE registry..... | 20 |
| Add a Language Pack and Change Locale | 20 |
| Appendices:..... | 21 |
| Appendix A: Using Vista Driverpacks from Driverpacks.net | 21 |
| Appendix B: Exclude files during capture using an ImageX configuration file..... | 21 |
| Exclusion List Guidelines | 21 |
| Appendix C: Booting Windows PE from USB Flash Drive (UFD)..... | 22 |
| Appendix D: Continual Reboots After Deploying Image | 23 |
| Appendix E: Mount WIM while booted to Windows PE | 23 |
| Additional Resources: | 24 |
| References: | 24 |
| Technical Advisors:..... | 24 |

Introduction:

With the release of Vista, Microsoft has developed a new method of deploying machines. Part of this new deployment method is the use of file based images called the Windows Imaging format or WIM.

What's the difference between a Ghost image and a WIM image?

Ghost doesn't know or care what format the drive or partition is that it is copying: FAT, FAT32, Linux EXT3, HPFS, or NTFS. To it disks are just partitions and partitions are sectors. Ghost will read a sector from the hard disk, compress it and place it into an image file, then continue with the next sector, and the next until it has finished copying the disk or partition, which is why it is called "sector-based" imaging.

WIMs are file- and folder- based, and deploy in a mostly non-destructive manner. When you deploy a sector-based image to a hard disk, you destroy whatever partitions sit on that disk; re-imaging a system with Ghost means wiping out the drive. However, when deploying a WIM to a system it means only overwriting whatever files are in the WIM, and no others. As an example, if you had a Windows XP workstation whose hard disk contained a folder named "c:\saveddocs" which held hundreds of documents and you then re-apply the original image to that computer, then you would have a clean copy of the operating system because you overwrote the old one, but the c:\saveddocs folder would still exist in its untouched state.

You can also use the WIM format at no cost. All of the WIM tools are a free download from Microsoft as part of the Windows Automated Installation Kit. Ghost, Acronis and other products like them in general all cost money.

~~One other important difference is that as of this writing there is currently no method to multicast a WIM image to multiple machines at once. This is being added to Server 2008 with the Windows Deployment Services.~~ Multicast ability has now been added to Server 2008 r2, in order to use this option you need to be using PE v2.1 from Vista SP1.


Requirements:

To work through this tutorial you will need the following:

- Technician Machine to run utilities on with the following:
 - WAIK (Windows Automated Installation Kit) installed on Technician machine
 - Based on Vista (PE 2.0):
<http://www.microsoft.com/downloads/details.aspx?familyid=C7D4BC6D-15F3-4284-9123-679830D629F2&displaylang=en>
 - Based on Vista SP1 (PE 2.1):
<http://www.microsoft.com/downloads/details.aspx?FamilyId=94BB6E34-D890-4932-81A5-5B50C657DE08&displaylang=en>
 - Technician machine will need a CD/DVD-R/RW
 - Minimum 2 GB of free disk space to install WAIK and work on Windows PE image.
- Source system
 - System configured as desired and ready for image capture.
- Destination system
 - System to test the deployment of your image to.
 - System must be the same or compatible with the source system unless you are using the Universal Imaging Utility
- External USB drive or network file share with enough space to store the image, a secondary drive in the source system is also acceptable.
- Blank CD/DVD-R/RW media, number depends upon method of deployment and image size.

Creating a basic Windows PE disk for troubleshooting or image capture or deployment:

1. Download and install WAIK on technician system machine.
2. Create your source system with software installed and configured as desired and ready to be imaged as you have always done.
3. On technician system launch '**Windows PE Tools Command Prompt**' from '**Microsoft Windows AIK**' on the Programs Menu. The entire CD build process is performed in this CMD window.
(NOTE: If you are using Vista you will need to run as administrator)



```
c:\ Windows PE Tools Command Prompt
Updating path to include peimg, oscdimg, imagex
C:\Program Files\Windows AIK\Tools\PETools\
C:\Program Files\Windows AIK\Tools\PETools\..\x86

C:\Program Files\Windows AIK\Tools\PETools>_
```

4. To begin building a Windows PE CD type: ***copype.cmd x86 c:\winpe_x86***
 - a. Where **x86** is the desired system architecture (x86 or amd64)
 - b. And **c:\winpe_x86** is your desired working location for Windows PE; this directory must **not** already exist.
5. Now mount the Windows PE boot image as a folder on your machine so that it may be edited by typing the following:
imagex /mountrw c:\winpe_x86\winpe.wim 1 c:\winpe_x86\mount
6. To copy servicing tools into the mounted PE image, type:
xcopy "C:\Program Files\Windows AIK\Tools\Servicing" c:\winpe_x86\mount\Windows /s
xcopy "C:\Program Files\Windows AIK\Tools\x86" c:\winpe_x86\mount\Windows /s /Y
7. If you wish to add any additional files such as additional diagnostic tools, Ghost32 or other standalone applications copy them now to: (Remember this is a limited Windows environment it will not have all of the functionality of a normal Windows installation.)
c:\winpe_x86\mount – this will be the X: drive when booted from the CD and everything here loads into memory, the more you put here the more memory that will be required.
c:\winpe_x86\ISO – this is the CD itself, put large files like the image files here.

8. Install additional support options:

These are not required for image capture or deployment, but if you need the ability to run scripts, use WMI, or any other item listed below it must be explicitly installed before continuing. If you decide you want them later you will have to start the process over.

- a. Windows Recovery Environment component
peimg /install=WinPE-SRT-Package c:\winpe_x86\mount\Windows
 - b. HTML application Support
peimg /install=WinPE-HTA-Package c:\winpe_x86\mount\Windows
 - c. Microsoft Data Access Component support
peimg /install=WinPE-MDAC-Package c:\winpe_x86\mount\Windows
 - d. Windows Script Host support
peimg /install=WinPE-Scripting-Package c:\winpe_x86\mount\Windows
 - e. Microsoft XML (MSXML) parser support
peimg /install=WinPE-XML-Package c:\winpe_x86\mount\Windows
 - f. WMI (Windows Management Instrumentation)
peimg /install=WinPE-WMI-Package c:\winpe_x86\mount\windows
9. Prepare the Windows PE Image by typing, this removes unneeded items from the image:
peimg /prep c:\winpe_x86\mount\Windows /f
10. Install hardware drivers if necessary before continuing; see [Installing Drivers](#) for more info.
peimg /inf=<path to inf> c:\winpe_x86\mount\windows
11. To commit the changes made to Windows PE, type:
imagex /unmount c:\winpe_x86\mount /commit
12. Export the WIM to remove space held by files no longer in use:
imagex /boot /export c:\winpe_x86\winpe.wim 1 c:\winpe_x86\smallpe.wim
13. Copy the updated and compacted WIM to the ISO folder for creation:
copy c:\winpe_x86\smallpe.wim c:\winpe_x86\ISO\sources\boot.wim /Y
14. Create your ISO image:
oscdimg -n -bc:\winpe_x86\etfsboot.com c:\winpe_x86\ISO c:\winpe_x86\winpe_x86.iso
15. Using your preferred CD creation software burn the disk image **c:\winpe_x86\winpe_x86.iso** to CD or CD-RW.

Installing Drivers

Drivers can be added to both an online and offline images using different tools. Drvload loads drivers to a booted Windows PE environment while the PEimg tool installs them to the offline image. If the driver requires a reboot to activate you should install it using PEimg, Drvload will ignore the reboot but will not initialize the .sys driver.

Installing drivers to offline Windows PE image:

1. Collect the drivers you wish to install and save them to a folder based on the device they are designed for.
C:\drivers\LAN
C:\drivers\MSD
etc.
2. Extract the drivers from their compressed packages; this process will vary depending upon the method used by the manufacturer. Place each driver in its own sub-folder under its device type folder c:\drivers\net, c:\drivers\msd, etc.
c:\drivers\net\intel100
c:\drivers\net\broadcom100
c:\drivers\MSD\intel
c:\drivers\MSD\AMD
etc.
3. Run the following command for each folder, where **<source path>** is the directory containing the driver INF file, and **<destination path>** (i.e. c:\winpe_x86\mount) is the path to you mounted Windows PE image:
peimg /inf=<source path>*.inf <destination path>\windows

Installing drivers to running Windows PE:

1. Collect the drivers you wish to use after booting.
2. Drivers should be extracted to folder and placed in folder on CD or USB device.
3. Run the following command after booting from you PE CD where **<source path>** is the directory containing the desired driver(s), multiple **<source paths>** can be specified by separating them with a space:
Drvload <source path> <source path> <source path 3>

Capturing a WIM disk images using Windows PE and ImageX:

Capture to USB or local drive:

1. Connect your USB or secondary drive to the machine to be imaged.
2. Turn on computer and boot from your newly create Windows PE CD.
3. Windows PE will load and launch a CMD window and run WPEINIT, do not close this window as this will cause the machine to reboot.
4. Verify the USB/secondary drive and the drive to be imaged are both seen, the internal drive should be C:, and the USB or secondary drive D:, but this is not always the case depending upon BIOS configurations. The CD will be the next drive letter after the last partition and the RAM drive where the OS is running from will be X:. As long as you know what is on your drive a simple DIR listing of each drive should suffice to determine the drive letters. For our example we will assume the internal hard drive is C: and the USB/secondary drive is D:
5. To capture the first partition in the CMD window type:
imagex /capture c:\ d:\C_Image.wim "Drive C" "My image description" /compress maximum
Note: Imagex only captures or deploys partitions; it does not capture full disks. **If your disk contains more than one partition you will need to capture each individually.** You can capture each partition to the same file by using the following command to append the additional partitions to the existing WIM (assuming D: is the second partition and E: is the USB/secondary drive):
imagex /append d:\ e:\C_Image.wim "Drive D"
6. If you will be putting your image on CD/DVD you may need to split the final image into smaller pieces to fit on the media do that now with this command:
imagex /split d:\C_Image.wim 650
This will split the image into 650mb chunks on in the same location as the image file. You can split the image into larger or smaller size chunks by changing the 650 to you desired size in megabytes. If you wish to split it to a different location add the path after the source file name:
imagex /split d:\c_image.wim d:\splitfile 650
7. You are now ready to deploy your image.

Capture to a network file share:

You must have created your Windows PE CD with the appropriate network drivers in order to capture the image to the network.

1. Turn on computer and boot from your newly create Windows PE CD.
2. Windows PE will load and launch a CMD window and run WPEINIT, do not close this window as this will cause the machine to reboot.
3. Verify the drive to be imaged is seen, the internal drive should be C:. The CD will be the next drive letter after the last hard drive partition and the RAM drive where the OS is running from will be X:. As long as you know what is on your drive a simple DIR listing of each drive should suffice to determine the drive letters.
4. There are two options to connect to a network share:
 - a. Map share to a drive letter:
 - i. In CMD window type: ***NET USE K: \\myserver\images***
 - ii. The net use command in Windows PE is the same as in Windows XP.
 - b. Use UNC path:
 - i. In CMD window type: ***SETLOCAL ENABLEEXTENSIONS***
 - ii. You will now be able to use UNC paths in the CMD window in place of drive letters.
5. If you chose to use UNC paths replace **K:** with the path to your share (**\\myserver\images**) in the following steps.
6. To capture the first partition in the CMD window type:
imagex /capture c:\ K:\C_Image.wim "Drive C" "My image description" /compress maximum
Note: Imagex only captures or deploys partitions; it does not capture full disks. **If your disk contains more than one partition you will need to capture each individually.** You can capture each partition to the same file by using the following command to append the additional partitions to the existing WIM (assuming D: is the second partition and E: is the USB drive):
imagex /append K:\ e:\C_Image.wim "Drive D"
7. If you will be putting your image on CD/DVD you may need to split the final image into smaller pieces to fit on the media do that now with this command:
imagex /split K:\C_Image.wim 650
This will split the image into 650mb chunks on in the same location as the image file. You can split the image into larger or smaller size chunks by changing the 650 to you desired size in megabytes. If you wish to split it to a different location add the path after the source file name:
imagex /split K:\c_image.wim d:\splitfile 650
8. You are now ready to deploy your image.

Partition and Format the Hard Drive from Windows PE:

Unlike Ghost, ImageX does not partition and format the drive as the image is deployed. It will in fact overwrite only the files that are contained within the WIM file. To start off a machine from scratch with a new image you must first partition and format the drive as desired. Following is the procedure for partitioning and formatting the hard drive from Windows PE as a single partition.

If the image you are deploying does not contain SP3 see [Appendix D: Continual Reboots After Deploying Image](#).

Partition and Format drive from Windows PE Single Partition:

1. From CMD window run **DISKPART**. DISKPART will present its own prompt.



```
C:\WINDOWS\system32\cmd.exe - diskpart
C:\WINDOWS\system32>diskpart
Microsoft DiskPart version 5.1.3565
Copyright (C) 1999-2003 Microsoft Corporation.
On computer: LAPTOP1-XP
DISKPART> list disk

   Disk ###  Status   Size      Free      Dyn  Gpt
   -----  -
   Disk 0    Online   75 GB     0 B
   Disk 1    Online  233 GB     0 B
DISKPART>
```

2. Determine Disk order, type: **LIST DISK**
3. Select the disk to be imaged, change the number to correspond to your destination, type: **SELECT DISK 0**
4. To erase everything on the disk type: **CLEAN**
5. Create primary partition: **CREATE PART PRI**
6. Select the newly created partition: **SELECT PART 1**
7. Set partition active: **ACTIVE**
8. Assign it a drive letter: **ASSIGN LETTER=C:**
9. Format the drive: **FORMAT FS=NTFS LABEL="CDRIVE" QUICK**
10. Exit DISKPART: **EXIT**

Partition and Format drive from Windows PE Two Partitions:

1. From CMD window run **DISKPART**. DISKPART will present its own prompt.



```
C:\WINDOWS\system32\cmd.exe - diskpart
C:\WINDOWS\system32>diskpart
Microsoft DiskPart version 5.1.3565
Copyright (C) 1999-2003 Microsoft Corporation.
On computer: LAPTOP1-XP
DISKPART> list disk

   Disk ###  Status         Size           Free           Dyn  Gpt
   -----  -
   Disk 0    Online         75 GB          0 B            Dyn  Gpt
   Disk 1    Online        233 GB          0 B            Dyn  Gpt

DISKPART>
```

2. Determine Disk order, type: **LIST DISK**
3. Select the disk to be imaged, change the number to correspond to your destination, type: **SELECT DISK 0**
4. To erase everything on the disk type: **CLEAN**
5. Create primary partition, size in MB: **CREATE PART PRI SIZE=1000**
6. Select the newly created partition: **SELECT PART 1**
7. Set partition active: **ACTIVE**
8. Assign it a drive letter: **ASSIGN LETTER=C:**
9. Format the drive: **FORMAT FS=NTFS LABEL="CDRIVE" QUICK**
10. Create extended partition: **CREATE PART EXT**
11. Create logical disk: **CREATE PART LOG**
12. Select logical disk: **SELECT PART 2**
13. Format logical partition: **FORMAT FS=NTFS LABEL="CDRIVE" QUICK**
Note: Notice you do not have to assign a drive letter to a partition to format it. It formats the currently selected partition.
14. Assign it a drive letter: **ASSIGN LETTER=E**
Note: D: is already assigned to your CD, the letters assigned to the drive are good only for this session of Windows PE and have no bearing anywhere else.
15. Exit DISKPART: **EXIT**

DISKPART Scripting:

DISKPART is also scriptable so that you do not have to manually type in all the commands to partition your disk. **WARNING:** If your disks are not in the order expected when the script is run the wrong disk could be selected and wiped clean thereby losing data.

To run a DISKPART script use the following command: ***DISKPART /s myscript.txt***

SinglePartition.txt

```
SELECT DISK 0
CLEAN
CREATE PART PRI
SELECT PART 1
ACTIVE
ASSIGN LETTER=C:
FORMAT FS=NTFS LABEL="CDRIVE" QUICK
EXIT
```

DualPartition.txt

```
SELECT DISK 0
CLEAN
CREATE PART PRI SIZE=1000
SELECT PART 1
ACTIVE
ASSIGN LETTER=C:
FORMAT FS=NTFS LABEL="CDRIVE" QUICK
CREATE PART EXT
CREATE PART LOG
SELECT PART 2
ASSIGN LETTER=D
FORMAT FS=NTFS LABEL="DDRIVE" QUICK
EXIT
```

Deploying WIM images:

Deploy from USB drive:

1. Connect USB drive to destination machine.
2. Boot from Windows PE CD
3. Partition and format the partition as shown above.
4. Deploy image to drive:
imagex /apply d:\C_Image.wim "Drive C" C:
5. If you need to make any changes to the image you can now access all of the files written from the image and modify them as needed.
6. Remove media and reboot.

Deploy from Network Share:

You must have created your Windows PE CD with the appropriate network drivers in order to capture the image to the network.

1. Boot from Windows PE CD
2. Partition and format the partition as shown above.
3. There are two options to connect to a network share:
 - a. Map share to a drive letter:
 - i. In CMD window type: ***NET USE K: \\myserver\images***
 - ii. The net use command in Windows PE is the same as in Windows XP.
 - b. Use UNC path:
 - i. In CMD window type: ***SETLOCAL ENABLEEXTENSIONS***
 - ii. You will now be able to use UNC paths in the CMD window in place of drive letters.
4. If you chose to use UNC paths replace **K:** with the path to your share (**\\myserver\images**) in the following steps.
5. Deploy image to drive:
imagex /apply K:\C_Image.wim "Drive C" C:
6. If you need to make any changes to the image you can now access all of the files written from the image and modify them as needed.
7. Remove media and reboot.

Deploy from CD/DVD:

1. First, add the WIM image to your CD/DVD.
 - a. Repeat the process to create a Windows PE CD to the point where you can add your own custom files.
 - b. Add the image WIM file to the ISO folder.
 - c. Finish the ISO image creation and burn to CD/DVD.
2. Boot target machine from your Windows PE CD/DVD containing your image file.
3. Partition and format drive as needed.
4. Determine the drive letter of your CD/DVD.
5. Replace d:\ in `imagex /apply` command with the letter of your CD/DVD as needed to access WIM.
6. Apply image: ***imagex /apply d:\C_Image.wim "Drive C" C:***
7. If you need to make any changes to the image you can now access all of the files written from the image and modify them as needed.
8. Remove media and reboot.

Command Files for automation of Base Windows PE ISO Image:

If desired change the variable values at the beginning of these CMD scripts to point to the directories you wish to use. Place these CMD files in **C:\program files\Windows AIK\Tools\PETools** and run them from the **“Windows PE Tools Command Prompt”** or you need to add the path to the PETools directory to your Path statement which is done by `pesetenv.cmd` when you open the **“Windows PE Tools Command Prompt”**. Each of these command scripts can be run individually and you processes run in between them or if no other additions are being made to the Windows PE image the `Build_All.cmd` script can be run after setting the environment variables at the beginning of it to the desired settings.

Create_Mount.cmd

Creates initial Windows PE image, mounts the image and adds the Imagex servicing tools.

```
Set DestFolder=C:\winpe_x86
Set Arch=x86
copype.cmd %Arch% %DestFolder%
imagex /mountrw %DestFolder%\winpe.wim 1 %DestFolder%\mount
xcopy "C:\Program Files\Windows AIK\Tools\Servicing" %DestFolder%\mount\Windows /s
xcopy "C:\Program Files\Windows AIK\Tools\%Arch%" %DestFolder%\mount\Windows /s /Y
exit
```

Install_Packages.cmd

Installs additional packages to the image, and then preps the image.

```
Set DestFolder=C:\winpe_x86
rem peimg /install=WinPE-SRT-Package %DestFolder%\mount\Windows
peimg /install=WinPE-HTA-Package %DestFolder%\mount\Windows
peimg /install=WinPE-MDAC-Package %DestFolder%\mount\Windows
peimg /install=WinPE-Scripting-Package %DestFolder%\mount\Windows
peimg /install=WinPE-XML-Package %DestFolder%\mount\Windows
peimg /install=WinPE-WMI-Package %DestFolder%\mount\windows
peimg /prep %DestFolder%\mount\Windows /f
```

Install_Drivers.CMD

When calling pass a variable pointing to the driver directory which contains the drivers you wish to install, the given directory will be recursively searched. Usage example: **Install_drivers c:\drivers\LAN**

```
set Drivers=%1
set DestFolder=c:\winpe_x86
pushd %Drivers%
for /f %%i in ('dir /s /b *.inf') do (
    peimg /inf:%%i %DestFolder%\mount\windows
    if ERRORLEVEL 1 echo %%i>>%Drivers%\bad_driver.txt
)
popd
```

Commit_Changes.cmd

Commits the changes to the image and removes excess space.

```
Set DestFolder=C:\winpe_x86
imagex /unmount %DestFolder%\mount /commit
Imagex /boot /export c:\winpe_x86\winpe.wim 1 c:\winpe_x86\smallpe.wim
```

Create_ISO.CMD

Copy smaller boot.wim to ISO and create ISO.

```
Set DestFolder=C:\winpe_x86
copy %DestFolder%\smallpe.wim %DestFolder%\ISO\sources\boot.wim /Y
oscdimg -n -b%DestFolder%\etfsboot.com %DestFolder%\ISO %DestFolder%\winpe_x86.iso
```

Build_All.CMD

```
Set Architecture=x86
Set PEDest=c:\winpe_x86
Set ISOName=winpe_x86.iso
Set installmsd=true
Set MSDPath=c:\drivers\MSD
Set installlan=true
Set LanPath=c:\drivers\LAN
Set removeboot=true
Set installghost=true
Set GhostPath=C:\Program Files\Symantec\Ghost

pushd %cd%
REM Create and mount PE image
call cotype.cmd %Architecture% %PEDest%
imagex /mountrw %PEDest%\winpe.wim 1 %PEDest%\mount

REM Copy ImageX tools to image
xcopy "C:\Program Files\Windows AIK\Tools\Serviceing" %PEDest%\mount\Windows /s
xcopy "C:\Program Files\Windows AIK\Tools%\Architecture%" %PEDest%\mount\Windows /s /Y

REM Install additional Packages in image
REM Not Active - for future - peimg /install=WinPE-SRT-Package %PEDest%\mount\Windows
peimg /install=WinPE-HTA-Package %PEDest%\mount\Windows
peimg /install=WinPE-MDAC-Package %PEDest%\mount\Windows
peimg /install=WinPE-Scripting-Package %PEDest%\mount\Windows
peimg /install=WinPE-XML-Package %PEDest%\mount\Windows
peimg /install=WinPE-WMI-Package %PEDest%\mount\Windows
REM Install Drivers to image
if "%installmsd%"=="true" call Install_Drivers.cmd %MSDPath%
if "%installlan%"=="true" call Install_Drivers.cmd %LANPath%
peimg /prep %PEDest%\mount\Windows /f
```

```
REM Install 32-Bit Symantec Ghost tools
```

```
if "%installghost%"=="true" copy "%GhostPath%\*32.exe" %PEDest%\mount\windows
```

```
REM Unmount and cleanup image
```

```
imagex /unmount %PEDest%\mount /commit
```

```
imagex /boot /export %PEDest%\winpe.wim 1 %PEDest%\smallpe.wim
```

```
copy %PEDest%\smallpe.wim %PEDest%\ISO\sources\boot.wim /Y
```

```
REM Remove 'Press any key to boot message
```

```
if "%removeboot%"=="true" del /q %pedest%\iso\boot\bootfix.bin
```

```
REM Create ISO
```

```
oscdimg -n -b%PEDest%\etfsboot.com %PEDest%\ISO %PEDest%\%ISOName%
```

```
popd
```

Further Customizations:

Adding your own startup commands or applications

With the image mounted edit `c:\winpe_x86\mount\windows\system32\startnet.cmd` and add your own commands to the end of the file to automate functions at startup. Or you can create a text file named `c:\winpe_x86\mount\windows\system32\Winpeshl.ini`, set the **AppPath** value in `Winpeshl.ini` to the application you wish to run, command line options are not supported.

Use the **[LaunchApps]** section to run applications with command-line options. The applications run in the order listed; separate the name of the application from its options with a comma.

Winpeshl.ini

```
[LaunchApp]
AppPath=%systemdrive%\system32\myshell.exe
[LaunchApps]
%SYSTEMDRIVE%\mydir\application1.exe, -option1 -option2
application2.exe, -option1 -option2
```

Remove boot prompt

To remove the “**Press any Key to boot from CD or DVD**” prompt when booting from CD, delete **bootfix.bin** from `c:\winpe_x86\ISO\boot` before running `OSCDIMG` to create the CD ISO image.

Using Ghost instead of ImageX

You can use Ghost instead of ImageX to handle imaging during the building of your Windows PE ISO. At a minimum you will need to add `GHOST32.EXE` to your CD, you can also use the other 32-bit application included like `GDISK32`, `GhRegEdit32`, `GhWalk32`, `Ghostexp`, etc. Then you can use Ghost as you have in the past but use the 32-bit Windows version instead of the 16-bit DOS version.

The following will copy all of the 32-bit Ghost tools to the mounted image:

```
copy "c:\Program Files\Symantec\Ghost\*32.exe" c:\winpe_x86\mount\windows
copy "c:\Program Files\Symantec\Ghost\ghostexp.exe" c:\winpe_x86\mount\windows
copy "c:\Program Files\Symantec\Ghost\ghostsrv.exe" c:\winpe_x86\mount\windows
```

Modify Background Image

With the image mounted replace `c:\winpe_x86\mount\windows\winpe.bmp` with your preferred image.

Modify the WinPE registry

With the image is mounted:

1. Open Regedit on technician computer
2. Select HKEY_LOCAL_MACHINE
3. Load the PE registry hive:
File -> Load Hive
Find c:\winpe_x86\mount\windows\system32\config
Select the desired Hive:
 - a. Software – HKLM\Software
 - b. System – HKLM\System
 - c. Default – HKU\DEFAULT
 - d. Components – HKLM\Components
4. Name key: WINPE
5. Make desired changes within HKEY_LOCAL_MACHINE\WINPE
6. Select WINPE
7. Unload hive: File -> Unload Hive -> Confirm

Add a Language Pack and Change Locale

With the image is mounted, and before running imagex /prep:

1. If it is not already part of the PE image import your language pack:
peimg /import=<path> c:\winpe_x86\mount\windows
2. Install the language pack once imported:
peimg /install=<package> c:\winpe_x86\mount\windows
i.e.: peimg /install=*JA-JP* c:\winpe_x86\mount\windows
3. Set you desired language:
peimg /lang=<culture> c:\winpe_x86\mount\windows
i.e.: peimg /lang=JA-JP c:\winpe_x86\mount\windows

Changing Locale:

1. ***intlcfg -all:<language> -image:c:\winpe_x86\mount***
Where language is the desired locale/language you wish to use in the format ,language>-<REGION>, i.e. intlcfg –all:en-us –image:c:\winpe_x86\mount
4. Continue with further customizations, then run peimg /prep command.

Appendices:

Appendix A: Using Vista Driverpacks from Driverpacks.net

At the time of this writing the 10.03 is the current version of the LAN and MassStorage Driverpacks.

PE 2.1 (Vista SP1) does not contain drivers for the network adapter for VMWare Workstation nor did the Vista Driverpacks. You will need to extract them from the VMWare Tools CD and install them in your image. If you mount the Windows.ISO from the VMWare install folder the drivers are in <CD> \program files\VMware\VMware Tools\Drivers\vmxnet\win2k, install both INF files.

Appendix B: Exclude files during capture using an ImageX configuration file

By default ImageX excludes certain files from being included in the image during a capture.

Default Exclusion list:

```
[ExclusionList]
\$\ntfs.log
\hiberfil.sys
\pagefile.sys
"System Volume Information"
\RECYCLER
\Windows\CSC

[CompressionExclusionList]
*.mp3
*.zip
*.cab
\WINDOWS\inf\*.pnf
```

You can define your own additional exclusion by creating a configuration INI file and referencing it during image capture. To use your custom list add **/config x:<config file path>\configuration.ini** to your ImageX capture command:

```
imagex /config x:<config file path>\configuration.ini /capture c:\ d:\C_Image.wim "Drive C" "My image description" /compress maximum
```

If you name the configuration file **wimscript.ini** and place it in the same folder as ImageX it will run automatically when using the **/capture** switch without needing to use the **/config** switch.

Exclusion List Guidelines

- You can only use wildcards in the last component in a file path that does not begin with a preceding back slash. For example, **myfolder*.txt**
- You can use a preceding backward slash to limit file- and directory-matching relative to the root directory. For example, if you capture the "c:\" drive with the previous exclusion list,

\myfolder

\folder\subfolder

Then ImageX will exclude the following root directories and their files,

C:\myfolder

c:\folder\subfolder

But will not exclude sub-folders of the same name for example,

C:\main\myfolder

c:\data\folder\subfolder

- You can override the default exclusion list by using the **[ExclusionException]** section. For example,

[ExclusionException]

\pagefile.sys

"\System Volume Information"

- If an explicit **[ExclusionException]** section is provided in the WIM configuration file, it will always take precedence over the **[Exclusion List]** section.
- You cannot override the default compression exclusion list by using the **[ExclusionException]** section.

Appendix C: Booting Windows PE from USB Flash Drive (UFD)

Before you can place Windows PE on a UFD device, you must use Diskpart to format the UFD with a Windows Vista or Windows PE environment.

1. From a running Windows Vista operation system or Windows PE session, insert your UFD device.
2. At a command prompt, use Diskpart to format the device as NTFS spanning the entire device, and set the partition as active. For example,

```
Diskpart
select disk 1
clean
create partition primary
select partition 1
active
format fs=NTFS
assign
exit
```

The example above assumes disk 1 is the UFD.

3. On your technician computer, copy all the content in the \ISO directory onto your UFD device. You can manually create the directory structure or use the xcopy command to automatically build and copy the appropriate files from your technician computer to your UFD device. For example,

xcopy c:\winpe_x86\iso*.* /s /e /ff:

where c: is your technician computer hard disk and f: is your UFD device..

Appendix D: Continual Reboots After Deploying Image

If you find that after deploying your image it continuously reboots and you receive a Stop error:
Stop: 0x000000ED (0x81F85678, 0xC000014f, 0x00000000,0x00000000)
Unmountable_boot_volume

It is caused by the differences in the way the XP and Vista partition hard drives. To fix this you need to add the following DWORD registry settings to your Windows PE image in
HKEY_LOCAL_MACHINE\SYSTEM\CURRENTCONTROLSET\SERVICES\VDS\ALIGNMENT

LessThan4GB = 0
Between4_8GB = 0
Between8_32GB = 0
GreaterThan32GB = 0

See [Modify the WinPE registry](#) on how to modify the registry offline, this can also be added to the registry after booting from the CD and running Regedit.

Another option is to install SP3 to you XP installation you are capturing.

More information available at in [KB931760](#).

Appendix E: Mount WIM while booted to Windows PE

I've not come up with a reason to do this, but the process is included for completeness.

Add to StartNet.cmd or run manually once booted from CD (single line):

```
%systemRoot%\System32\rundll32.exe setupapi,InstallHinfSection DefaultInstall 132  
x:\windows\wimfltr.inf
```

Now you can use imagex with /mount or /mountw to mount your images to a folder.

Additional Resources:

- ImageX GUI interface –
<http://www.msfm.org/board/index.php?showtopic=99148>
- DISKPART GUI Interface – Disk Partitioner v1.2
<http://www.911cd.net/forums//index.php?showtopic=13516&hl=\disk+partitioner\+v1.2>
- Vista Driverpacks – Network and Mass Storage driver packs
<http://driverpacks.net/driverpacks/windows/vista/x86>
- Ghost Menu HTA for WinPE 2.x
<http://www.msfm.org/board/WinPE-20-simple-Ghost-Me-t95507.html>

References:

- <http://technet2.microsoft.com/WindowsVista/en/library/129a1712-e3d8-46c1-bc09-a14349dc67db1033.mspx?mfr=true>
- WINPE.CHM – Help file installed by WAIK

Technical Advisors:

- JuMz - msfn.org
- Chiners_68 – msfn.org
- p4ntboy – msfn.org
- DigeratiPrime – msfn.org
- Tripredacus – msfn.org